



# Équations et programme FORTRAN de l'Indice Forêt-Météo de la méthode canadienne



**SERVICE CANADIEN DES FORÊTS**  
**GOUVERNEMENT DU CANADA**

Le Service canadien des forêts (SCF) réunit la majorité des spécialistes fédéraux en foresterie. Son objectif général est de promouvoir l'aménagement et l'utilisation judicieux des ressources forestières du Canada pour le plus grand bien économique, social et environnemental des Canadiens.

Voici les principales fonctions du SCF:

1. Coordonner les politiques fédérales afin de favoriser l'amélioration de la gestion des ressources et l'expansion de l'industrie forestière.
2. Fournir une orientation scientifique et technologique dans le domaine de la foresterie, par la recherche et le développement.
3. Fournir et analyser les statistiques et l'information nationales et internationales qui serviront à établir les politiques.
4. Mettre au point et homologuer des codes et des normes en matière de rendement des produits du bois.
5. Protéger les forêts canadiennes en luttant contre les ravageurs étrangers.
6. Parrainer l'utilisation éventuelle des ressources forestières pour la production d'énergie.
7. Adhérer aux objectifs environnementaux du Service canadien des forêts.

Divers organismes fédéraux participent aux programmes forestiers, et un comité de la stratégie forestière fédérale a été créé pour coordonner les activités fédérales en matière de foresterie. Le SCF a été désigné organisme directeur.

Le Service canadien des forêts comprend une administration centrale, six centres de recherches forestières et deux instituts nationaux. Les centres de recherches forestières doivent répondre aux impératifs régionaux et entretenir une liaison étroite avec les ministères provinciaux des Forêts. Ils participent également à des programmes nationaux dont ils assument fréquemment la direction. Les instituts nationaux sont les foyers des programmes d'envergure nationale.

# **Équations et programme FORTRAN de l'Indice Forêt-Météo de la méthode canadienne**

---

**C.E. Van Wagner et T.L. Pickett  
Institut forestier national de Petawawa**

©Ministre des Approvisionnements et Services Canada, 1985

ISBN 0-662-93118-1

N° de catalogue Fo64-33/1985F

Des exemplaires de cette publication peuvent être obtenus  
sans frais à l'adresse suivante:

Environnement Canada  
Centre de distribution  
151, Jean-Proulx  
Ottawa (Ontario)  
K1A 1C7

This publication is also available in English under the title  
*Equations and FORTRAN Program for the Canadian  
Forest Fire Weather Index System.*

## Table des matières

---

	Page
Résumé.....	IV
Abstract.....	IV
Introduction.....	1
Symboles utilisés dans les équations.....	3
Équations et méthodes de calcul.....	5
Description du programme.....	9
Symboles utilisés dans le programme.....	11
Programme F-32F.....	12
Exemple de données d'entrée.....	16
Exemple de résultats.....	17
Ouvrages de références.....	18



## RÉSUMÉ

---

Ce rapport présente, en complément de l'Indice Forêt-Météo de la méthode canadienne, des équations connues qu'on a retouchées. Ces précisions mathématiques veulent rendre l'indice du combustible léger plus rationnel et davantage compatible avec les nouveaux développements que connaît la Méthode canadienne d'évaluation des dangers d'incendie de forêt. Il s'agit de changements mineurs qui ne devraient pas entraîner de problème dans l'utilisation de la méthode. Le rapport présente également un programme FORTRAN qui se veut une base pour traiter les équations dans le plus pur langage mathématique.

## ABSTRACT

---

Improved official equations are presented for the 1984 version of the Canadian Forest Fire Weather Index System. The most recent mathematical refinements serve to further rationalize the Fine Fuel Moisture Code and render it more compatible with other developments in the Canadian Forest Fire Danger Rating System. The effect of these changes is so slight that no problems are anticipated in converting from the previous version to this new one. Also given is a FORTRAN program intended as a standard for processing the equations in their most accurate mathematical form.

## Introduction

---

Le présent rapport accompagne la quatrième édition des tables de l'Indice Forêt-Météo de la méthode canadienne (Service canadien des forêts, 1984). Il présente les équations sur lesquelles est fondée cette nouvelle édition, ainsi qu'un programme permettant le calcul de l'Indice Forêt-Météo (IFM). Ce programme est écrit en FORTRAN 77, pour être exécuté sur un ordinateur DEC PDP-11/44. Il est possible de l'adapter à un grand nombre d'ordinateurs plus petits.

Ce rapport remplace le rapport d'information PS-X-58 (Van Wagner et Pickett, 1975<sup>1</sup>) qui présentait les équations de la version 1976 de l'IFM. Van Wagner (1974) a décrit la structure de l'IFM ainsi que la façon dont cet indice a été élaboré dans la publication n° 1333 du Service canadien des forêts, mais les calculs utilisés dans ce rapport sont en partie caducs. Une nouvelle version de cette publication sera bientôt éditée.

La version 1976 de l'IFM comporte les changements mathématiques suivants par rapport à la version originale de 1970:

1. Changement dans l'équation donnant la teneur en eau à l'équilibre (TEE) pendant une phase de mouillage dans l'indice du combustible léger (ICL), pour éliminer une anomalie lorsque l'humidité relative (HR) est très élevée.
2. Remplacement de la correction de température originale dans l'ICL par deux corrections distinctes sur la TEE et la vitesse de dessèchement.
3. Augmentation de la quantité de pluie éliminée dans l'indice de sécheresse (IS) de 1,524 mm à 2,8 mm.
4. Modification du facteur combustible disponible,  $f(D)$ , aux valeurs supérieures à 80 de l'indice du combustible disponible (ICD), afin de permettre à l'IFM de se stabiliser en cas d'extrême sécheresse.

Nota: D'autres acronymes et symboles sont définis à l'annexe 1.

Les autres changements mathématiques apparaissant dans la présente version rationalisent davantage la structure de l'ICL. Cet indice s'est révélé adéquat, mais des modifications apportées à la Méthode canadienne d'évaluation des dangers d'incendie de forêt (MCEDIF), dont l'IFM fait partie, ont rendu nécessaires a) une conversion réaliste de la valeur de l'indice à la teneur en eau réelle du combustible, b) une base commune pour l'ICL journalier normal et l'indice horaire sur 24 heures (Van Wagner 1977) et c) une possibilité de modifier rapidement les calculs dans les cas particuliers. En conséquence, les changements suivants ont été apportés à l'ICL:

1. Une nouvelle échelle d'humidité, l'échelle FF, avec une teneur en eau maximale plus élevée (250 % au lieu de 101 %).
2. Une nouvelle équation sur l'effet de la pluie basée sur la teneur en eau plutôt que sur l'indice.
3. Une vitesse de dessèchement corrigée basée sur la teneur en eau exprimée par la nouvelle échelle d'humidité, mais qui ne change pas le taux de variation de la valeur de l'indice.
4. Une vitesse de mouillage variable lorsque la TEE est supérieure à la teneur en eau initiale, à la place de la vitesse fixe utilisée auparavant.
5. Une correction du facteur combustible léger,  $f(F)$ , dans l'indice de propagation initiale (IPI), pour qu'il y ait correspondance avec la nouvelle échelle de conversion de l'humidité.

---

<sup>1</sup>C.E. Van Wagner est chercheur et T.L. Pickett programmeur à l'Institut forestier national de Petawawa, dont l'adresse est: Chalk River (Ontario) K0J 1J0.

Ces modifications n'entraînent que très peu de différences dans les résultats du système d'équations. Cela signifie que les variations sur les indices obtenus seront suffisamment faibles pour qu'on puisse continuer d'utiliser sans interruption les valeurs journalières des indices et pour que la continuité statistique des données annuelles soit préservée.

Comme auparavant, le programme permet le calcul des indices d'une saison entière en une seule fois, mais pour une station seulement. Son but principal est de montrer comment programmer les équations de façon à obtenir les résultats dans la forme mathématique la plus pure possible, sans contraintes ou limites artificielles autres que celles qu'impose la nature des équations. Il vise à servir d'étalon auquel on pourra comparer les résultats d'autres programmes.

La version 1984 (comme la version précédente) comprend le calcul de l'indice journalier de sévérité (IJS) tel que décrit par Van Wagner (1970). L'IJS est une fonction de l'IFM définie spécialement pour le calcul de moyennes sur une période choisie en un endroit donné ou dans une zone donnée à partir d'un nombre quelconque de stations. On considère que l'IFM, ne convient pas pour le calcul des moyennes et qu'il doit être utilisé seulement comme valeur journalière. La moyenne de l'IJS pour toute une saison, par exemple, serait appelé l'indice saisonnier de sévérité (ISS) et servirait à comparer les conditions météo-incendie d'une année à l'autre ou d'un endroit à l'autre.



## Symboles utilisés dans les équations

Toutes les grandeurs utilisées dans les équations numérotées sont représentées dans la liste qui suit par des lettres uniques, quelquefois accompagnées d'un indice. Les symboles sont regroupés suivant leur place dans l'ensemble. Toutes les teneurs en eau sont des pourcentages calculés en fonction de la masse sèche. Midi est l'heure normale.

### Météo

$T$	– température à midi, °C
$H$	– humidité relative à midi, %
$W$	– vitesse du vent à midi, km/h
$r_o$	– précipitation mesurée tous les jours à midi dans un endroit dégagé, mm d'eau
$r_f$	– précipitation réelle, ICL
$r_e$	– précipitation réelle, IH
$r_d$	– précipitation réelle, IS

### Indice du combustible léger (ICL)

$m_o$	– teneur en eau du combustible léger le jour précédent
$m_r$	– teneur en eau du combustible léger après la pluie
$m$	– teneur en eau du combustible léger après dessèchement
$E_d$	– TEE du combustible léger pour le dessèchement
$E_w$	– TEE du combustible léger pour le mouillage
$k_o$	– valeur intermédiaire dans le calcul de $k_d$
$k_d$	– vitesse de dessèchement logarithmique, ICL, $\log_{10}m/\text{jour}$
$k_l$	– valeur intermédiaire dans le calcul de $k_w$
$k_w$	– vitesse de mouillage logarithmique, $\log_{10}m$ par jour
$F_o$	– ICL de la veille
$F$	– ICL

### Indice de l'humus (IH)

$M_o$	– teneur en eau de l'humus le jour précédent
$M_r$	– teneur en eau de l'humus après la pluie
$M$	– teneur en eau de l'humus après le dessèchement
$K$	– vitesse de dessèchement logarithmique, IH, $\log_{10}m/\text{jour}$
$L_e$	– longueur réelle de la journée dans IH, heures
$b$	– facteur pente dans l'effet de la pluie sur IH
$P_o$	– IH de la veille
$P_r$	– IH après la pluie
$P$	– IH

**Indice de sécheresse (IS)**

Q	– humidité équivalente de IS, multiples de 0,254 mm
Q <sub>o</sub>	– humidité équivalente de l'IS de la veille
Q <sub>r</sub>	– humidité équivalente après la pluie
V	– évapotranspiration potentielle, multiples de 0,254 mm d'eau par jour
L <sub>r</sub>	– coefficient de la longueur du jour dans IS
D <sub>o</sub>	– IS de la veille
D <sub>r</sub>	– IS après la pluie
D	– IS

**Indice de comportement de l'incendie (IPI, ICD, IFM)**

f(W)	– facteur vent
f(F)	– facteur humidité du combustible léger
f(D)	– facteur humidité de l'humus
R	– indice de propagation initiale (IPI)
U	– indice du combustible disponible (ICD)
B	– IFM (forme intermédiaire)
S	– IFM (forme finale)

**Indice de sévérité**

IJS	– indice journalier de sévérité
-----	---------------------------------

## Équations et méthodes de calcul

### Indice du combustible léger (ICL)

$m_o = 147,2 (101 - F_o) / (59,5 + F_o)$	(1)
$r_f = r_o - 0,5,$	$r_o > 0,5$ (2)
$m_r = m_o + 42,5 r_f (e^{-100/(251 - m_o)})(1 - e^{-6,93/r_f}),$	$m_o \leq 150$ (3a)
$m_r = m_o + 42,5 r_f (e^{-100/(251 - m_o)})(1 - e^{-6,93/r_f}) + 0,0015 (m_o - 150)^2 r_f^{0,5},$	$m_o > 150$ (3b)
$E_d = 0,942 H^{0,679} + 11e^{(H-100)/10} + 0,18 (21,1 - T)(1 - e^{-0,115H})$	(4)
$E_w = 0,618 H^{0,753} + 10e^{(H-100)/10} + 0,18 (21,1 - T)(1 - e^{-0,115H})$	(5)
$k_o = 0,424 [1 - (H/100)^{1,7}] + 0,0694 W^{0,5} [1 - (H/100)^8]$	(6a)
$k_d = k_o \times 0,581 e^{0,0365T}$	(6b)
$k_l = 0,424 \left[ 1 - \left( \frac{100 - H}{100} \right)^{1,7} \right] + 0,0694 W^{0,5} \left[ 1 - \left( \frac{100 - H}{100} \right)^8 \right]$	(7a)
$k_w = k_l \times 0,581 e^{0,0365T}$	(7b)
$m = E_d + (m_o - E_d) \times 10^{-k_d}$	(8)
$m = E_w - (E_w - m_o) \times 10^{-k_w}$	(9)
$F = 59,5 (250 - m) / (147,2 + m)$	(10)

L'ICL se calcule de la façon suivante:

1. F de la veille devient  $F_o$ .
2. Calculer  $m_o$  en fonction de  $F_o$  à l'aide de l'équation 1.
- 3a. Si  $r_o > 0,5$ , calculer  $r_f$  à l'aide de l'équation 2.
- b. Calculer  $m_r$  en fonction de  $r_f$  et  $m_o$  à l'aide de l'équation 3a ou 3b.
  - (i) si  $m_o \leq 150$ , utiliser l'équation 3a.
  - (ii) si  $m_o > 150$ , utiliser l'équation 3b.
- c.  $m_r$  devient alors le nouveau  $m_o$ .
4. Calculer  $E_d$  à l'aide de l'équation 4.
- 5a. Si  $m_o > E_d$ , calculer  $k_d$  à l'aide des équations 6a et 6b.
- b. Calculer  $m$  à l'aide de l'équation 8.
6. Si  $m_o < E_d$ , calculer  $E_w$  à l'aide de l'équation 5.
- 7a. Si  $m_o < E_w$ , calculer  $k_w$  à l'aide des équations 7a et 7b.
- b. Calculer  $m$  à l'aide de l'équation 9.
8. Si  $E_d \geq m_o \geq E_w$ , prendre  $m = m_o$ .
9. Calculer F en fonction de m à l'aide de l'équation 10. On obtient l'ICL de la journée.

L'utilisation de ces équations comporte deux restrictions:

- L'équation 3 (a ou b) ne s'applique pas lorsque  $r_o \leq 0,5$  mm; cela signifie que, par temps sec, le sous-programme précipitation doit être omis.
- m a une limite supérieure de 250; cela signifie que lorsque l'équation 3 (a ou b) donne  $m_r > 250$ , il faut prendre  $m_r = 250$ .

# Indice de l'humus (IH)

$r_e = 0,92r_o - 1,27,$	$r_o > 1,5$ (11)
$M_o = 20 + e^{(5,6348 - P_o/43,43)}$	(12)
$b = 100/(0,5 + 0,3 P_o),$	$P_o \leq 33$ (13a)
$b = 14 - 1,3 \ln P_o,$	$33 < P_o \leq 65$ (13b)
$b = 6,2 \ln P_o - 17,2,$	$P_o > 65$ (13c)
$M_r = M_o + 1000r_e/(48,77 + br_e)$	(14)
$P_r = 244,72 - 43,43 \ln (M_r - 20)$	(15)
$K = 1,894 (T + 1,1) (100 - H) L_e \times 10^{-6}$	(16)
$P = P_o \text{ (ou } P_r) + 100K$	(17)

L'IH se calcule de la façon suivante:

1. P de la veille devient  $P_o$ .
- 2a. Si  $r_o > 1,5$ , calculer  $r_e$  à l'aide de l'équation 11.  
b. Calculer  $M_o$  en fonction de  $P_o$  à l'aide de l'équation 12.  
c. Calculer b à l'aide de l'équation appropriée parmi 13a, 13b ou 13c.  
d. Calculer  $M_r$  à l'aide de l'équation 14.  
e. Calculer  $P_r$  en fonction de  $M_r$  à l'aide de l'équation 15.  $P_r$  devient le nouveau  $P_o$ .
3. Prendre  $L_e$  dans le tableau 1 ci-dessous.
4. Calculer K à l'aide de l'équation 16.
5. Calculer P en fonction de  $P_o$  (ou  $P_r$ ) à l'aide de l'équation 17. C'est l'IH de la journée.

L'utilisation des équations donnant IH comporte trois restrictions:

- Les équations 11 à 15 ne s'appliquent pas à moins que  $r_o > 1,5$ ; cela signifie qu'il faut omettre le sous-programme précipitation par temps sec.
- Théoriquement,  $P_r$  ne peut pas être inférieur à zéro. Si on obtient des valeurs négatives à l'étape 2e, il faut prendre zéro.
- On ne doit pas utiliser des valeurs de T inférieures à  $-1,1$  dans l'équation 15. Si  $T < -1,1$ , il faut prendre  $T = -1,1$ .

**Tableau 1. Longueur réelle de la journée ( $L_e$ ) pour le calcul de IH**

Mois:	Janv.	Févr.	Mars	Avr.	Mai	Juin	Juill.	Août	Sept.	Oct.	Nov.	Déc.
$L_e$ :	6,5	7,5	9,0	12,8	13,9	13,9	12,4	10,9	9,4	8,0	7,0	6,0



### Indice de sécheresse (IS)

$r_d = 0,83r_o - 1,27,$	$r > 2,8$ (18)
$Q_o = 800e^{-D_o/400}$	(19)
$Q_r = Q_o + 3,937r_d$	(20)
$D_r = 400 \ln(800/Q_r)$	(21)
$V = 0,36 (T + 2,8) + L_f$	(22)
$D = D_o \text{ (ou } D_r) + 0,5V$	(23)

IS se calcule de la façon suivante:

1. D de la veille devient  $D_o$ .
  - 2a. Si  $r_o > 2,8$ , calculer  $r_d$  à l'aide de l'équation 18.  
 b. Calculer  $Q_o$  en fonction de  $D_o$  à l'aide de l'équation 19.  
 c. Calculer  $Q_r$  à l'aide de l'équation 20.  
 d. Calculer  $D_r$  en fonction de  $Q_r$  à l'aide de l'équation 21.  $D_r$  devient le nouveau  $D_o$ .
  3. Prendre  $L_f$  dans le tableau 2 ci-dessous.
  4. Calculer V à l'aide de l'équation 22.
  5. Calculer D en fonction de  $D_o$  (ou  $D_r$ ) à l'aide de l'équation 23. C'est l'IS de la journée.
- L'utilisation des équations donnant IS comporte quatre restrictions:
- Les équations 18 à 21 ne s'appliquent pas à moins que  $r_o > 2,8$ ; cela signifie qu'on doit omettre le sous-programme précipitation par temps sec.
  - Théoriquement,  $D_r$  ne peut être inférieur à zéro. Si on obtient des valeurs négatives à l'étape 2d, il faut prendre zéro.
  - On ne doit pas utiliser des valeurs de T inférieures à  $-2,8$ , dans l'équation 22. Si  $T < -2,8$ , prendre  $T = -2,8$ .
  - V ne peut pas être négatif. Si l'équation 22 donne un résultat négatif, prendre  $V = 0$ .

**Tableau 2. Coefficient de la longueur de la journée ( $L_f$ ) pour le calcul de IS.**

Mois:	Janv.	Févr.	Mars	Avr.	Mai	Juin	Juill.	Août	Sept.	Oct.	Nov.	Déc.
$L_f$ :	-1,6	-1,6	-1,6	0,9	3,8	5,8	6,4	5,0	2,4	0,4	-1,6	-1,6

### Indice de propagation initiale (IPI)

$f(W) = e^{0,05039W}$	(24)
$f(F) = 91,9e^{-0,1386m} [1 + m^{5,31}/(4,93 \times 10^7)]$	(25)
$R = 0,208 f(W) f(F)$	(26)

### Indice du combustible disponible (ICD)

$U = 0,8 PD/(P + 0,4D),$	$P \leq 0,4D$ (27a)
$U = P - [1 - 0,8D/(P + 0,4D)][0,92 + (0,0114P)^{1,7}],$	$P > 0,4D$ (27b)

**Indice Forêt-Météo (IFM)**

$f(D) = 0,626U^{0,809} + 2,$	$U \leq 80$ (28a)
$f(D) = 1000/(25 + 108,64 e^{-0,023U}),$	$U > 80$ (28b)
$B = 0,1 R f(D)$	(29)
$\ln S = 2,72 (0,434 \ln B)^{0,647},$	$B > 1$ (30a)
$S = B,$	$B \leq 1$ (30b)

IPI, ICD et IFM se calculent de la façon suivante:

1. Calculer  $f(W)$  et  $f(F)$  à l'aide des équations 24 et 25.
2. Calculer  $R$  à l'aide de l'équation 26. C'est l'IPI de la journée.
3. Calculer  $U$  à l'aide de l'équation 27a si  $P \leq 0,4D$  ou à l'aide de l'équation 27b si  $P > 0,4D$ . C'est l'ICD de la journée.
4. Calculer  $f(D)$  à l'aide de l'équation 28a si  $U \leq 80$ . Si  $U > 80$ , utiliser l'équation 28b.
5. Calculer  $B$  à partir de l'équation 29.
6. Si  $B > 1$ , calculer  $S$  d'après son logarithme donné par l'équation 30a. Si  $B \leq 1$ , prendre  $S = B$  selon l'équation 30b.  $S$  est l'IFM de la journée.

**Indice journalier de sévérité (IJS)**

$IJS = 0,0272 (IFM)^{1,77}$	(31)
-----------------------------	------

## Description du programme

---

Le programme décrit dans le présent rapport est issu d'un programme écrit par Simard (1970) de l'ancien Institut de recherches sur les feux de forêt. Ce programme était conçu pour traiter des données météorologiques collectées dans un grand nombre de stations pendant une saison complète. Le remplacement de données manquantes était prévu, ainsi que des limites dans le cas de données météorologiques anormales. Il s'agissait d'un programme complexe nécessitant à l'époque un ordinateur puissant.

Engisch et Walker (1971) ont simplifié le programme de Simard de façon à ce qu'il traite les données vérifiées relevées à une seule station pendant une saison, éliminant toutes les limites superflues. Kean (1975) a ensuite révisé ce programme, introduisant plusieurs modifications mathématiques qui avaient fait l'objet d'études récentes. Cette version est cependant devenue caduque à la suite de la décision de convertir les mesures au système métrique. Le rapport d'information PS-X-58 (Van Wagner et Pickett, 1975) décrit la version métrique du programme de Kean. Le programme faisant l'objet du présent rapport intègre les changements mathématiques décrits plus haut, mais il s'exécute de la même façon que le programme présenté dans le rapport PS-X-58.

Voici comment s'exécute le programme. On introduit d'abord le numéro du mois de départ et le nombre de jours de ce mois pendant lesquels des données sont collectées. Par exemple, si on commence le 13 avril, on collecte des données pendant 18 jours dans le mois n° 4, et la donnée d'entrée est 418. On introduit ensuite les observations météorologiques journalières dans l'ordre suivant: température (°C), humidité relative (%), vent (km/h) et pluie (mm). La température et la quantité de pluie sont données avec un seul chiffre décimal, l'humidité relative et la vitesse du vent sont des nombres entiers. On peut introduire les données météorologiques en unités anglaises, mais elles sont converties en unités métriques dans les résultats.

Les valeurs initiales normales des trois indices d'humidité du combustible, à savoir  $ICL = 85$ ,  $IH = 6$  et  $IS = 15$ , sont incluses dans le programme, mais il est possible d'introduire n'importe quelle autre valeur initiale désirée.

Les résultats sont donnés sur douze colonnes: la date (mois et jour), les quatre types de données météorologiques, les trois indices d'humidité du combustible ( $ICL$ ,  $IH$ ,  $IS$ ), les trois indices de comportement de l'incendie ( $IPI$ ,  $ICD$ ,  $IFM$ ), et l'indice journalier de sévérité ( $IJS$ ). L'ordinateur s'arrête automatiquement après le dernier jour. Le programme n'inclut pas de limites sur les données d'entrée autres que celles nécessaires pour éviter des anomalies mathématiques. Tous les indices d'humidité sont reportés d'un jour à l'autre en virgule flottante, et il en va de même pour toutes les quantités intermédiaires utilisées dans le calcul de l' $IFM$  journalier. Le résultat est donc dans la forme mathématique la plus pure possible et il constitue une référence à laquelle on peut comparer les résultats obtenus avec d'autres programmes ou avec les tables de l' $IFM$  de la méthode canadienne (Service canadien des forêts 1984).

Ce programme<sup>1</sup> standard sert principalement d'outil pour des travaux de développement sur le système de l' $IFM$  et de référence, comme on l'a déjà exposé. Il est évident que des programmes opérationnels conçus pour traiter des observations météorologiques effectuées quotidiennement dans un grand nombre de stations pendant toute une saison d'incendies auront des données d'entrée et de sortie différentes.

Deux points méritent un commentaire. Le premier est le degré de précision nécessaire lors de l'impression des codes et indices journaliers. Le programme standard arrondit tous les codes et indices au dixième et les imprime avec un chiffre décimal. Cependant, des résultats entiers conviennent sans aucun doute dans la plupart des cas. Quelquefois, le premier chiffre décimal peut être souhaitable pour l' $ICL$  (surtout au-dessus de 90) et l' $IPI$  (surtout au-dessous de 10). L' $IJS$  est donné avec deux décimales, ce qui convient le mieux.

---

<sup>1</sup>Programme F-32 (30 août 1984). Pour tout renseignement, s'adresser à l'Institut forestier national de Petawawa, Service canadien des forêts, Chalk River (Ontario) K0J 1J0.



Le second concerne la précision avec laquelle les trois indices d'humidité (ICL, IH et IS) sont reportés d'un jour à l'autre. Le programme standard reporte les valeurs journalières des indices en virgule flottante, ce qui équivaut à environ sept chiffres significatifs (quatre ou cinq décimales). En pratique, cependant, si la capacité de mémoire de l'ordinateur est un facteur limitant, on peut dans certains cas accepter des résultats avec une précision imparfaite. Des essais ont été effectués en reportant les trois indices d'humidité d'un jour à l'autre avec trois degrés réduits de précision: 1. avec deux décimales, 2. avec une décimale et 3. arrondis au nombre entier. Les résultats des essais sur deux saisons, exprimés en pourcentage de jours-erreurs pour chaque indice, sont donnés au tableau 3.

**Tableau 3. Pourcentages de jours pour lesquels on observe un écart entre les indices d'humidité arrondis au nombre entier obtenus à trois niveaux de précision du report des indices d'un jour à l'autre et les indices obtenus avec un report normal en virgule flottante.**

Indice	Report avec deux décimales	Report avec une décimale	Report d'un nombre entier			
	% de jours avec un écart de 1	% de jours avec un écart de 1	1	% de jours avec un écart de		
				2	3	1 ou plus
ICL	—	1,9	6,2	—	—	6,2
IH	0,5	4,3	41,8	4,9	—	46,7
IS	8,1	5,2	41,3	1,4	1,6	44,3
IPI	—	0,5	3,3	—	—	3,3
ICD	—	3,5	36,1	8,2	1,4	45,7
IFM	—	0,5	9,5	—	—	9,5

Le tableau 3 permet de tirer les conclusions suivantes:

- Le report des indices d'humidité avec deux décimales donne des résultats d'une exactitude presque parfaite, sauf pour IS qui présente des écarts occasionnels de 1 unité, mais ces écarts sont sans grandes conséquences.
- Dans certains cas pratiques, les reports des indices avec une décimale peut être suffisamment précis. Une erreur de 1 se produirait environ 1 jour sur 25 dans le cas de l'IH, environ 1 jour sur 50 dans le cas de l'ICL et environ 1 jour sur 200 dans le cas de l'IFM.
- Le report des indices d'humidité arrondis au nombre entier donne des résultats généralement inexacts et n'est pas recommandé.

Dans le tableau 3, sur les six indices, IS est le seul qui présente des anomalies. Du point de vue mathématique, IS peut difficilement s'autocorriger, si l'on compare avec ICL et IH, ce qui explique sa tendance à s'écarter d'une unité et à rester ainsi les jours suivants. Un échantillonnage beaucoup plus important serait nécessaire pour établir un comportement moyen de IS.

Le programme standard de calcul de l'Indice Forêt-Météo est présenté aux pages suivantes, avec des exemples de données d'entrée et de résultats.



## Symboles utilisés dans le programme

### Météo

T	– température, °C
H	– humidité relative, %
W	– vent, km/h
R,RA	– pluie, mm

### Indice du combustible léger (ICL)

FO	– ICL initial ou de la veille
WMO	– teneur en eau initiale
WM	– teneur en eau finale du jour
ED	– TEE pour le dessèchement
EW	– TEE pour le mouillage
Z	– valeur intermédiaire de X
X	– vitesse logarithmique de dessèchement ou de mouillage
FFM	– ICL du jour

### Indice de l'humus (IH)

PO	– IH initial ou de la veille
RK	– facteur de dessèchement
EL(J)	– longueur réelle de la journée pour le mois J
RW	– pluie réelle
PR	– IH après la pluie
WMI	– teneur en eau initiale
WMR	– teneur en eau après la pluie
B	– facteur pluie

### Indice de sécheresse (IS)

DOT	– IS initial ou de la veille
PE	– facteur de dessèchement
FL(J)	– coefficient de la longueur de la journée pour le mois J
RW	– pluie réelle
SMI	– humidité équivalente de IS
DR	– IS après la pluie

### Indices de comportement de l'incendie (IPI, ICD, IFM)

FM	– teneur en eau du combustible léger du jour
SF	– facteur humidité du combustible léger
SI	– IPI
P	– rapport permettant de corriger ICD lorsqu'il est inférieur à IH
CC	– facteur IH permettant de corriger ICD lorsqu'il est inférieur à IH
BB	– forme intermédiaire de IFM
SL	– logarithme du IFM final
DSR	– indice journalier de sévérité

## Programme F-32F

```
C      NO. DU PROGRAMME: F-32F
C
C      VERSION NORMALISEE DE 1984 DU SYSTEME DE L'INDICE FORET-METEO
C      DE LA METHODE CANADIENNE. ETABLIE LE 30 AOUT 1984.
C
C      EN FORTRAN 77 AUX FINS DU DEC-PDP-11/44 A L'IFNP.
C
C      PERMET LA LECTURE DES DONNEES EN UNITES ANGLAISES OU METRIQUES,
C      MAIS SEULES LES UNITES METRIQUES SONT IMPRIMEES.
C
C      EXPRIME LES INDICES DE SORTIES A UNE DECIMALE, SAUF L'INDICE
C      JOURNALIER DE SEVERITE, QUI SORT A DEUX DECIMALES.
C
C      LMON = LONGUEUR DES MOIS
C      EL = FACTEURS DE LONGUEUR DES JOURNEES DE L'IH
C      FL = FACTEURS DE LONGUEUR DES JOURNEES DE L'IS
C
C      DIMENSION LMON(12), EL(12), FL(12),AST(2),TITLE(20)
C      LOGICAL*1 DAT(9),YES,YES1,ANS,ANS1
C      LOGICAL *1 INFMT(40)
C      TYPE 5
C      5 FORMAT(' INTRODUIRE LE NOM DU FICHIER: PAR EX., DONNEES
C      * D'' ECHANTILLONNAGE ', $)
C      ACCEPT 10, INFMT
C      10 FORMAT(40A1)
C      INFMT(40)=0
C      OPEN(UNIT=1,NAME=INFMT,TYPE='OLD',READONLY)
C      OPEN(UNIT=2,STATUS='NEW',NAME='F32FOUT.DAT')
C      DATA LMON /31,28,31,30,31,30,31,31,30,31,30,31/
C      DATA EL /6.5,7.5,9.0,12.8,13.9,13.9,12.4,10.9,9.4,8.0,7.0,
C      *6.0/
C      DATA FL /-1.6,-1.6,-1.6,.9,3.8,5.8,6.4,5.0,2.4,.4,-1.6,-1.6/
C      DATA AST /' ','*' /
C      DATA ANS /'0' /
C      TYPE 15
C      15 FORMAT('//X,' INTRODUIRE AU CLAVIER 0 OU 1 '//
C      *X'.....0 SI LES DONNEES SONT EXPRIMEES EN UNITES METRIQUES'/
C      *X' OU 1 SI LES DONNEES SONT EXPRIMEES EN UNITES ANGLAISES')
C      READ(5,*) IUNIT
C
C      20 - PRESENTATION EN UNITES METRIQUES
C      25 - PRESENTATION EN UNITES ANGLAISES
C
C      20 FORMAT(F4.1,2I4,F5.1)
C      25 FORMAT(F4.0,2I4,F5.2)
C
C      LECTURE PAR STATION ET PAR ANNEE
C
C      READ(1,30) TITLE
C      30 FORMAT(20A4)
C
C      CETTE SECTION PERMET DES VALEURS INITIALES FACULTATIVES DE L'ICL,
C      DE L'IH ET DE L'IS.
C
C      F0=85.0
C      P0=6.0
C      DOT=15.0
C      TYPE 35
C      35 FORMAT(' ICL=85.0, IH=6.0, IS=15.0; DESIREZ-VOUS UTILISER CES
C      * VALEURS NORMALES INITIALES? [O/N] '$)
C      ACCEPT 10,ANS1
C      IF(ANS1.EQ.ANS) GO TO 55
```

```

TYPE 40
40 FORMAT(' FFMC (F4.1): ', $)
ACCEPT 20, F0
TYPE 45
45 FORMAT(' DMC (F4.1): ', $)
ACCEPT 20, P0
TYPE 50
50 FORMAT(' DC (F4.1): ', $)
ACCEPT 20, DOT
C
C LECTURE DU MOIS DE DEPART ET DU NOMBRE DE JOURS QU'IL COMPREND
C
55 READ(1, 60) M, NDAYS
60 FORMAT(I1, I2)
WRITE(2, 65)
65 FORMAT(1H1'NO. DU PROGRAMME: F-32F')
CALL DATE(DAT)
WRITE(2, 70) DAT, TITLE
70 FORMAT(1X, 9A1///1X, 20A4//)
WRITE(2, 75) F0, P0, DOT
75 FORMAT(' VALEURS INITIALES DE L''ICL:', F5.1, ', DE L''IH:', F5.1,
*, ET DE L''IS:', F5.1, //)
DO 290 J=M, 12
NN=LMON(J)
IF(J.EQ.M) GO TO 80
IDAYS=1
GO TO 85
80 IDAYS=LMON(J)-NDAYS+1
C
C LECTURE DES DONNEES METEOROLOGIQUES QUOTIDIENNES
C
85 IAST=1
DO 290 I=IDAYS, NN
IF(IUNIT.EQ.0) GO TO 90
READ(1, 25, END=295) T, IH, IW, R
W=IW
GO TO 95
90 READ(1, 20, END=295) T, IH, IW, R
W=IW
95 IF(IUNIT.EQ.0) GO TO 100
T=(T-32.)*5./9.
W=W*1.609
R=R*25.4
100 TX=T
H=IH
RAIN=R
IF(IAST.GT.1) GO TO 110
WRITE(2, 105)
105 FORMAT(///, 1X, ' DATE TEMP HR VENT PLUIE
* ICL IH IS
* IPI ICD IFM IJS '//)
C
C INDICE DU COMBUSTIBLE LEGER
C
110 WMO=(147.2*(101-F0))/(59.5+F0)
IF(R.GT.0.5) GO TO 115
GO TO 125
115 RA=R-.5
IF(WMO.GT.150.) GO TO 120
WMO=WMO+42.5*RA*EXP(-100./(251.-WMO))*(1.-EXP(-6.93/RA))
GO TO 125
120 WMO=(WMO+42.5*RA*EXP(-100./(251.-WMO))*(1.-EXP(-6.93/RA)))+
*(0.0015*(WMO-150.))**2)*RA**5

```

```

125 IF(WMO.GT.250.) WMO=250.
    ED=0.942*(H**0.679)+(11.*EXP((H-100.)/10.))+0.18*(21.1-T)
    ** (1.-1./EXP(0.115*H))
    IF(WMO-ED) 130,135,140
130 EW=.618*(H**.753)+(10.*EXP((H-100.)/10.))+.18*(21.1-T)
    ** (1.-1./EXP(0.115*H))
    IF(WMO.LT.EW) GO TO 145
135 WM=WMO
    GO TO 150
140 Z=0.424*(1.-(H/100.))**1.7)+(0.0694*(W**0.5))*(1.-(H/100.))**8)
    X=Z*(0.581*(EXP(0.0365*T)))
    WM=ED+(WMO-ED)/10.**X
    GO TO 150
145 Z=.424*(1.-((100.-H)/100.))**1.7)+(0.0694*(W**.5))*(1.-((100.-H)
    */100.))**8)
    X=Z*(.581*(EXP(.0365*T)))
    WM=EW-(EW-WMO)/10.**X
150 FFM=(59.5*(250.-WM))/(147.2+WM)
    IF(FFM.GT.101.) GO TO 155
    IF(FFM) 160,165,165
155 FFM=101.
    GO TO 165
160 FFM=0.0

C
C   INDICE DE L'HUMUS
C
165 IF(T+1.1.GE.0.) GO TO 170
    T=-1.1
170 RK=1.894*(T+1.1)*(100.-H)*(EL(J)*0.0001)
175 IF(R.GT.1.5) GO TO 180
    PR=P0
    GO TO 205
180 RA=R
    RW=0.92*RA-1.27
    WMI=20.0+280./EXP(0.023*P0)
    IF(P0.LE.33.) GO TO 185
    IF(P0-65.) 190,190,195
185 B=100./(0.5+0.3*P0)
    GO TO 200
190 B=14.-1.3*ALOG(P0)
    GO TO 200
195 B=6.2*ALOG(P0)-17.2
200 WMR=WMI+(1000.*RW)/(48.77+B*RW)
    PR=43.43*(5.6348-ALOG(WMR-20.))
205 IF(PR.GE.0.) GO TO 210
    PR=0.0
210 DMC=PR+RK

C
C   INDICE DE SECHERESSE
C
    IF(T+2.8.GE.0.) GO TO 215
    T=-2.8
215 PE=(.36*(T+2.8)+FL(J))/2.
    IF(R.LE.2.8) GO TO 225
    RA=R
    RW=0.83*RA-1.27
    SMI=800.*EXP(-DOT/400.)
    DR=DOT-400.*ALOG(1.+((3.937*RW)/SMI))
    IF(DR.GT.0.) GO TO 220
    DR=0.0
220 DC=DR+PE
    GO TO 230
225 DR=DOT

```



```

      GO TO 220
230 IF(DC.GE.0.) GO TO 235
      DC=0.0
C
C      INDICE DE PROPAGATION INITIALE, INDICE DU COMBUSTIBLE DISPONIBLE,
C      INDICE FORET-METEO
C
235 FM=(147.2*(101.-FFM))/(59.5+FFM)
      SF=19.115*EXP(FM*(-.1386))*(1.+(FM**5.31)/4.93E07)
      SI=SF*EXP(0.05039*W)
240 IF(DMC.EQ.0.0.AND.DC.EQ.0.0) GO TO 245
      BUI=(0.8*DC*DMC)/(DMC+0.4*DC)
      GO TO 250
245 BUI=0.
250 IF(BUI.GE.DMC) GO TO 255
      P=(DMC-BUI)/DMC
      CC=0.92+(0.0114*DMC)**1.7
      BUI=DMC-(CC*P)
      IF(BUI.LT.0.) BUI=0.
255 IF(BUI.GT.80.) GO TO 260
      BB=0.1*SI*(0.626*BUI**0.809+2.)
      GO TO 265
260 BB=0.1*SI*(1000./((25.+108.64/EXP(0.023*BUI))))
265 IF(BB-1.0.LE.0.) GO TO 270
      SL=2.72*(0.434*ALOG(BB))**0.647
      FWI=EXP(SL)
      GO TO 275
270 FWI=BB
275 DSR = 0.0272*FWI**1.77
280 WRITE(2,285) J,I,TX,IH,IW,RAIN,FFM,DMC,DC,SI,BUI,FWI,DSR
285 FORMAT(1X,2I3,F6.1,I4,I6,F7.1,F7.1,F6.1,F7.1,3F6.1,F8.2)
      FO=FFM
      PO=DMC
      DOT=DC
      IAST=IAST+1
290 CONTINUE
295 WRITE(5,300)
300 FORMAT(/1X'RESULTATS SONT DANS F32FOUT.DAT ET
      * DOIVENT ETRE TRANSMIS AU TRAITEMENT'/
      *' EN DIFFERE A L'' IMPRIMANTE LIGNE PAR LIGNE.'//)
305 STOP
      END

```

## Exemple de données d'entrée

> DONNEES DE L'ESSAI D'IFM DU 13 AVRIL AU 31 MAI  
 418 ————— 4 représente le numéro du mois initial, ici avril  
 18 représente le nombre de jours dans le mois initial.

17.	42	25		
20.	21	25	2.4	
8.5	40	17		
6.5	25	6		
13.	34	24		
6.0	40	22	0.4	
5.5	52	6		
8.5	46	16		
9.5	54	20		
7.0	93	14	9.0	
6.5	71	17	1.0	
6.0	59	17		
13.	52	4		
15.5	40	11		
23.	25	9		
19.	46	16		
18.	41	20		
14.5	51	16		
14.5	69	11		
15.5	42	8		
21.	37	8		
23.	32	16		
23.	32	14		
27.	33	12		
28.	17	27		
23.5	54	20		
16.	50	22	12.2	
11.5	58	20		
16.	54	16		
21.5	37	9		
14.	61	22	0.2	
15.	30	27		
20.	23	11		
14.	95	3	16.4	
20.	53	4	2.8	
19.5	30	16		
25.5	51	20	6.0	
10.	38	24		
19.	27	16		
26.	46	11	4.2	
30.	38	22		
25.5	67	19	12.6	
12.	53	28	11.8	
21.	38	8		
13.	70	20	3.8	
9.	78	24	1.4	
11.	54	16		
15.5	39	9		
18.	36	5		

— 13 avril      — 30 avril  
 — 1<sup>er</sup> mai      — 31 mai

Température (°C)    Humidité relative (%)    Vent (km/h)    Pluie (mm)

## Exemple de résultats

NO. DU PROGRAMME: F-32F  
15-FEB-85

DONNEES DE L'ESSAI D'IFM DU 13 AVRIL AU 31 MAI

VALEURS INITIALES DE L'ICL: 85.0, DE L'IH: 6.0, ET DE L'IS: 15.0

DATE	TEMP	HR	VENT	PLUIE	ICL	IH	IS	IPI	ICD	IFM	IJS
4 13	17.0	42	25	0.0	87.7	8.5	19.0	10.9	8.5	10.1	1.63
4 14	20.0	21	25	2.4	86.2	10.4	23.6	8.8	10.4	9.3	1.40
4 15	8.5	40	17	0.0	87.0	11.8	26.1	6.5	11.7	7.6	0.98
4 16	6.5	25	6	0.0	88.8	13.2	28.2	4.9	13.1	6.2	0.68
4 17	13.0	34	24	0.0	89.1	15.4	31.5	12.6	15.3	14.8	3.22
4 18	6.0	40	22	0.4	88.7	16.5	33.5	10.7	16.4	13.5	2.71
4 19	5.5	52	6	0.0	87.4	17.2	35.4	4.0	17.1	5.9	0.62
4 20	8.5	46	16	0.0	87.4	18.5	37.9	6.6	18.4	9.7	1.50
4 21	9.5	54	20	0.0	86.8	19.7	40.6	7.4	19.6	11.0	1.89
4 22	7.0	93	14	9.0	29.9	10.1	29.5	0.0	10.9	0.0	0.00
4 23	6.5	71	17	1.0	49.4	10.7	31.6	0.4	11.6	0.2	0.00
4 24	6.0	59	17	0.0	67.3	11.4	33.7	1.3	12.3	0.9	0.02
4 25	13.0	52	4	0.0	77.8	13.0	37.0	1.1	13.9	0.8	0.02
4 26	15.5	40	11	0.0	85.5	15.4	40.7	3.9	15.9	5.5	0.56
4 27	23.0	25	9	0.0	91.5	19.8	45.8	8.4	19.8	12.2	2.27
4 28	19.0	46	16	0.0	89.9	22.5	50.2	9.5	22.4	14.3	3.02
4 29	18.0	41	20	0.0	90.0	25.2	54.4	11.7	25.1	17.7	4.40
4 30	14.5	51	16	0.0	88.4	27.0	57.9	7.7	27.0	13.3	2.65
5 1	14.5	69	11	0.0	85.7	28.3	63.0	4.0	28.2	8.0	1.08
5 2	15.5	42	8	0.0	87.4	30.8	68.2	4.4	30.8	9.1	1.35
5 3	21.0	37	8	0.0	89.4	34.5	74.3	5.9	34.4	12.3	2.30
5 4	23.0	32	16	0.0	91.0	38.8	80.9	11.1	38.7	21.0	5.96
5 5	23.0	32	14	0.0	91.2	43.1	87.4	10.3	43.0	21.1	5.99
5 6	27.0	33	12	0.0	91.7	48.1	94.7	9.9	47.9	21.7	6.30
5 7	28.0	17	27	0.0	95.2	54.5	102.1	34.5	54.3	52.6	30.20
5 8	23.5	54	20	0.0	89.7	57.4	108.8	11.3	57.2	25.9	8.61
5 9	16.0	50	22	12.2	62.2	29.9	91.8	1.4	33.0	3.0	0.19
5 10	11.5	58	20	0.0	76.7	31.3	96.3	2.3	34.5	5.4	0.55
5 11	16.0	54	16	0.0	83.5	33.4	101.6	3.8	36.7	8.9	1.31
5 12	21.5	37	9	0.0	88.7	37.1	107.8	5.6	39.9	12.8	2.46
5 13	14.0	61	22	0.2	86.7	38.7	112.8	8.1	41.6	17.3	4.24
5 14	15.0	30	27	0.0	89.6	41.7	117.9	15.9	44.2	28.8	10.44
5 15	20.0	23	11	0.0	92.1	45.9	123.9	10.1	47.7	21.9	6.40
5 16	14.0	95	3	16.4	21.3	20.1	97.0	0.0	26.5	0.0	0.00
5 17	20.0	53	4	2.8	51.0	18.3	103.0	0.2	25.3	0.2	0.00
5 18	19.5	30	16	0.0	82.3	22.1	108.9	3.3	29.3	6.8	0.81
5 19	25.5	51	20	6.0	75.4	16.4	106.4	2.1	23.7	3.8	0.29
5 20	10.0	38	24	0.0	84.3	18.2	110.6	6.4	25.8	11.3	1.99
5 21	19.0	27	16	0.0	90.3	22.1	116.4	10.0	29.9	17.2	4.19
5 22	26.0	46	11	4.2	77.6	18.7	117.7	1.6	26.8	2.9	0.18
5 23	30.0	38	22	0.0	90.2	23.8	125.5	13.4	32.3	22.0	6.49
5 24	25.5	67	19	12.6	65.3	13.1	108.5	1.4	20.2	1.9	0.08
5 25	12.0	53	28	11.8	55.4	7.7	91.6	1.2	12.8	0.8	0.02
5 26	21.0	38	8	0.0	80.8	11.3	97.8	1.9	17.6	2.6	0.14
5 27	13.0	70	20	3.8	61.7	8.4	97.9	1.2	13.8	0.9	0.02
5 28	9.0	78	24	1.4	64.5	9.0	101.9	1.7	14.7	2.0	0.09
5 29	11.0	54	16	0.0	77.6	10.5	106.3	2.0	16.8	2.8	0.17
5 30	15.5	39	9	0.0	85.4	13.1	111.5	3.5	20.3	5.8	0.61
5 31	18.0	36	5	0.0	88.5	16.3	117.1	4.4	24.2	7.9	1.06

## Ouvrages de références

---

- Engisch, R.L.; Walker, J.D. 1971. PDP-8L version of Simard's Fire Weather Index Program. Environ. Can., Can. For. Serv., Petawawa Forest Exp. Sta., Intern. Rep. PS-23. 10 p.
- Kean, W.A. 1975. A PDP-8L program for calculating the Fire Weather Index. Environ. Can., Can. For. Serv., Petawawa Forest Exp. Sta., Inf. Rep. PS-X-57. 12 p.
- Service canadien des forêts. 1984. Tables de l'Indice Forêt-Météo de la méthode canadienne. Environ. Can., rapp. tech. de for. n° 25 (4<sup>e</sup> éd.).
- Simard, A.J. 1970. Computer program to calculate the Canadian Forest Fire Weather Index. Environ. Can., Can. For. Serv., Forest Fire Res. Inst., Intern. Rep. FF-12. 18 p.
- Van Wagner, C.E. 1970. Conversion of Williams' severity rating for use with the Fire Weather Index. Environ. Can., Can. For. Serv., Petawawa Forest Exp. Sta., Inf. Rep. PS-X-21. 5 p.
- Van Wagner, C.E. 1974. Structure de l'indice canadien forêt-météo. Environ. Can., Serv. can. des for. publ. n° 1333. 44 p.
- Van Wagner, C.E. 1977. A method of computing fine fuel moisture content throughout the diurnal cycle. Environ. Can., Can. For. Serv., Petawawa Forest Exp. Sta. Info. Rep. PS-X-69. 15 p.
- Van Wagner, C.E.; Pickett, T.L. 1975. Equations and Fortran IV program for the 1976 metric version of the Forest Fire Weather Index. Can. For. Serv., Petawawa Forest Exp. Sta., Inf. Rep. PS-X-58. 20 p.





